

Supplementary Material

Bayesian model description

Let y_i be an indicator of whether individual i tests positive (i.e. equal to 1 if positive, 0 if negative). Furthermore, let π_i denote the probability the individual is seropositive and p_i denote the probability the individual's test outcome is positive, i.e. $\Pr(y_i = 1)$. Given a perfect diagnostic test, $p_i = \pi_i$, however there is non-zero probability of obtaining a false positive and false negative result the test and therefore these quantities are not equal. If se is the true sensitivity and sp is the true specificity of the test, the probabilities are related via

$$p_i = \pi_i * se + (1 - \pi_i) * (1 - sp). \quad (1)$$

Recognizing that seroprevalence may vary by sex and age, we consider the following logistic regression model:

$$\pi_i = \text{logit}^{-1}(\beta_1 + \alpha_{ag[i]}^{ag} + \alpha_{hh[i]}^{hh})$$

where $ag[i]$ indexes which age/gender subpopulation person i belongs, and $hh[i]$ indexes which household person i belongs.

As in Ref. [1], we place a uniform(0,1) prior distribution on the probability that an average person is positive by specifying a unit logistic prior for the intercept β_1 . The effects for age/gender and household have hierarchical priors

$$\alpha_j^{ag} \sim \text{normal}(0, \sigma^{ag})$$

$$\alpha_k^{hh} \sim \text{normal}(0, \sigma^{hh})$$

where σ^{ag} and σ^{hh} are modeled using a $\text{normal}^+(0, 0.5)$.

The model in Eq. (1) requires knowledge of the true sensitivity and specificity of the test, however this is never known in practice. Rather, these test parameters are themselves estimated from data and it is critical the uncertainty in the test characteristics be propagated through to prevalence estimates for accurate inference [2, 1]. We therefore expand our model to include the lab validation data reported by Roche for the Elecsys® Anti-SARS-CoV-2 assay.

Consider n_{se} positive control samples of which y_{se} correctly test positive and n_{sp} negative control samples of which y_{sp} correctly test negative. We model these lab validation results as binomial outcomes:

$$y_{se} \sim \text{binomial}(n_{se}, se),$$

$$y_{sp} \sim \text{binomial}(n_{sp}, sp),$$

and specify uniform(0,1) priors on the sensitivity se and specificity sp parameters.

For the purposes of this study, we considered $n_{se} = 88$, $y_{se} = 81$, $n_{sp} = 5272$, and $y_{sp} = 5262$ using all negative control samples and all positive samples from individuals at least seven days post a positive PCR test [3].

The Bayesian model was fit in R using Stan via the RStan package [4]. The model was fit separately to each area and wave. Posterior mean and 95% equal-tailed credible

intervals were reported for the model parameters based on four chains, each run for 5,000 iterations, with 5,000 iterations of burn-in. The (improved) \hat{R} Markov chain Monte Carlo diagnostic, described in Ref. [5], was less than 1.05 for all parameters, and all effective sample sizes were greater than 5,000. The Stan code for the model was adapted based on that given in Ref. [1] and is provided in full below.

To obtain seroprevalence estimates for entire regions, the age and gender prevalence estimates from the Bayesian model were post-stratified using demographic information specific to each area.

Bayesian model fitting: Stan code

```
// Fits model with separate effect for each age/sex group,
// while accounting for household dependence
data {
  int<lower = 0> N; // number of tests in the sample
  int<lower = 0, upper = 1> y[N]; // test outcomes (1 if positive, 0 if negative)
  int<lower = 0> N_hh; // number of households
  int<lower = 0> N_ag; // number of age/gender bins
  int<lower = 1, upper = N_hh> hh[N]; // household index
  int<lower = 1, upper = N_ag> ag[N]; // age/gender index
  int<lower = 0> y_spec; // count of negative outcomes
  int<lower = 0> n_spec; // number of negative controls
  int<lower = 0> y_sens; // count of positive outcomes
  int<lower = 0> n_sens; // number of positive controls
  int<lower = 0> J; // number of subpopulation cells
  vector<lower = 0>[J] N_pop; // subpopulation sizes for post-stratification
  real<lower = 0> coef_prior_scale; // .5
}
parameters {
  real b; // intercept
  real<lower = 0, upper = 1> sens;
  real<lower = 0, upper = 1> spec;
  real<lower = 0> sigma_hh;
  real<lower = 0> sigma_ag;
  vector<multiplier = sigma_hh>[N_hh] a_hh; // varying intercepts for households
  vector<multiplier = sigma_ag>[N_ag] a_ag; // varying intercepts for age/gender groups
}
model {
  vector[N] p = inv_logit(b + a_hh[hh] + a_ag[ag]);
  vector[N] p_sample = p * sens + (1 - p) * (1 - spec);
  y ~ bernoulli(p_sample);
  y_spec ~ binomial(n_spec, spec);
  y_sens ~ binomial(n_sens, sens);

  a_hh ~ normal(0, sigma_hh);
  a_ag ~ normal(0, sigma_ag);

  sens ~ beta(1,1);
  spec ~ beta(1,1);

  // prior on intercept
  b ~ logistic(0, 1);
}
```

```

sigma_hh ~ normal(0, coef_prior_scale);
sigma_ag ~ normal(0, coef_prior_scale);
}
generated quantities {
  real p_avg;
  vector[J] p_pop; // prevalence in the J post-stratification cells
  int count;
  count = 1;
  for (i_ag in 1:N_ag) {
    p_pop[count] = inv_logit(b + a_ag[i_ag]);
    count += 1;
  }
  p_avg = sum(N_pop .* p_pop) / sum(N_pop);
}

```

Seroprevalence estimates by age and gender




Figure 1: Posterior density estimates of seroprevalence based on the second survey for each age and gender subpopulation in each district.

References

- [1] Andrew Gelman and Bob Carpenter. Bayesian analysis of tests with unknown specificity and sensitivity. *medRxiv*, 2020.
- [2] Daniel B Larremore, Bailey K Fosdick, Sam Zhang, and Yonatan H Grad. Jointly modeling prevalence, sensitivity and specificity for optimal sample allocation. *bioRxiv*, 2020.

Gender	Age	District East	District Malir
Female	0-4	0.139 (0.048, 0.251)	0.080 (0.029, 0.143)
Female	5-9	0.128 (0.048, 0.221)	0.088 (0.040, 0.155)
Female	10-18	0.130 (0.057, 0.217)	0.086 (0.041, 0.143)
Female	19-39	0.149 (0.079, 0.238)	0.101 (0.054, 0.166)
Female	40-59	0.155 (0.078, 0.261)	0.084 (0.037, 0.142)
Female	60+	0.149 (0.056, 0.278)	0.099 (0.046, 0.197)
Male	0-4	0.139 (0.049, 0.258)	0.079 (0.027, 0.140)
Male	5-9	0.128 (0.039, 0.231)	0.082 (0.035, 0.141)
Male	10-18	0.170 (0.090, 0.284)	0.098 (0.049, 0.176)
Male	19-39	0.153 (0.080, 0.249)	0.079 (0.034, 0.133)
Male	40-59	0.217 (0.108, 0.406)	0.083 (0.033, 0.145)
Male	60+	0.145 (0.051, 0.268)	0.087 (0.035, 0.159)

Table 1: Seroprevalence posterior mean and 95% equal-tailed credible interval estimates based on the second survey for both districts.

- [3] Roche Diagnostics. Elecsys anti-sars-cov-2 package insert 2020-04. *VI. 0*, 422:423, 2020.
- [4] Stan Development Team et al. Rstan: the r interface to stan. *r* package version 2.17.3, 2018.
- [5] Aki Vehtari, Andrew Gelman, Daniel Simpson, Bob Carpenter, and Paul-Christian Bürkner. Rank-normalization, folding, and localization: An improved \hat{R} for assessing convergence of mcmc, 2019.